

Praktický pohled na programování systémů v reálném čase

Petr Holub

`hopet@ics.muni.cz`



Laboratoř pokročilých síťových technologií

IA158
2014-04-30

Přehled přednášky

Ada

Podpora paralelismu na úrovni jazyka

Systémy real-time

- Plánování a spouštění vláken
- Řazení do front
- Časovače a události
- Komunikace mezi vlákny

Omezující profily

Ada

- Ada 83/95/2005
- jazyk orientovaný na spolehlivé aplikace: vojáci, letectví, ...
- WORM: write once read many
- silně typovaný jazyk
- podpora paralelismu
- enkapsulace, výjimky, objekty, správa paměti, atd.
- GNAT
 - volně dostupný překladač, frontend k GCC
- materiály na webu
 - (Annotated) Reference Manual
<http://www.adaic.org/standards/ada05.html>
 - <http://stwww.weizmann.ac.il/g-cs/benari/books/index.html#ase>
 - <http://en.wikibooks.org/wiki/Programming:Ada>
 - <http://www.adahome.com/Tutorials/Lovelace/lovelace.htm>
 - <http://www.pegasoft.ca/resources/boblap/book.html>

Na co se podívat

- Struktura balíků, enkapsulace, pojmenování souborů
- Typový systém Ada, typy a podtypy, ukazatele (`access` VS. `access all`), generika, objekty, atributy proměnných

- Správa paměti
- Generické kontejnery
- Volby kompilace:

```
gnatmake hello.adb
```

VS.

```
gnatmake -gnatya -gnatyb -gnatyc -gnatye -gnatyf -gnatyi  
-gnatyk -gnatyl -gnatyn -gnatyp -gnatyr -gnatyt -g -gnato  
-gnatf -fstack-check hello.adb
```

Co v Adě neuděláte

... tedy co v ní neuděláte (*takhle*).

```
1 if(c = ntoa(b)) {}
```

Co v Adě neuděláte

... tedy co v ní neuděláte (*takhle*).

```
1 send(to, from, count)
  register short *to, *from;
3 register count;
  {
5     register n = (count + 7) / 8;
     switch(count % 8) {
7         case 0: do { *to = *from++;
9         case 7:     *to = *from++;
1        case 6:     *to = *from++;
12       case 5:     *to = *from++;
14       case 4:     *to = *from++;
16       case 3:     *to = *from++;
18       case 2:     *to = *from++;
20       case 1:     *to = *from++;
22             } while(--n > 0);
     }
  }
```

Zdroj: https://en.wikipedia.org/wiki/Duff%27s_device

Kopírování pole do memory-mapped výstupu, nikoli kopie pole.

Co v Adě neuděláte

... tedy co v ní neuděláte (*takhle*).

```

1  if(c = ntoa(b)) {}
3  #define _ -F<00||--F-OO--;
   int F=00,OO=00;main(){F_OO();printf("%1.3f\n",4.*-F/OO/OO);}F_OO()
5  {
7      -
8      -
9      -
10     -
11     -
12     -
13     -
14     -
15     -
16     -
17     -
18     -
19     -
20     -
21     -
   }

```

Zdroj: <http://www0.us.ioccc.org/1988/westley.c>

Protected Types – Monitory

- Implementace monitorů
 - funkce – nemohou měnit data
 - procedury – mohou měnit data
 - entry – strážžený vstup, mohou měnit data
 - efektivní paralelizace: podobne `ReadWriteLocku` v Javě
 - ◆ funkce mohou přistupovat paralelně
 - ◆ procedury a entries musí pracovat exkluzivně

```
protected type Muj_Typ is
2   procedure Nastav_hodnotu (n : Integer);
   procedure Odnastav_hodnotu;
4   function Zjistihodnotu return Integer;
   entry Pockej_na_nastaveni (n : Integer);
6 private
   Hodnota : Integer;
8   Nastaveno : Boolean := False;
end Muj_Typ;
```

Protected Types – monitory

```
10 protected body Muj_Typ is
11     procedure Nastav_hodnotu (n : Integer) is
12     begin
13         Hodnota := n;
14         Nastaveno := True;
15     end Nastav_hodnotu;
16
17     procedure Odnastav_hodnotu is
18     begin
19         Nastaveno := False;
20     end Odnastav_hodnotu;
21
22     function Zjistihodnotu return Integer is
23     begin
24         return Hodnota;
25     end Zjistihodnotu;
26
27     entry Pockej_na_nastaveni
28     when Nastaveno is
29     begin
30         null;
31     end Pockej_na_nastaveni;
32 end Muj_Typ;
```

Guarded Entries

- chránění dle privátního stavu

```
1  protected type Chraneno_Stavem is
2     entry Vstup;
3  private
4     I : Integer;
5  end Chraneno_Stavem;
6
7  protected body Chraneno_Stavem is
8     entry Vstup when I > 0 is
9     begin
10        null;
11     end Vstup;
12 end Chraneno_Stavem;
```

- používat pouze privátní proměnné
- např. implementace mutexů a semaforů

Guarded Entries

- Chránění dle atributů

```

1  protected type Chraneno_Stavem is
2      entry Vstup;
3  private
4      I : Integer;
5  end Chraneno_Stavem;

7  protected body Chraneno_Stavem is
8      entry Vstup when Vstup'Count > 4 is
9          begin
10             null;
11         end Vstup;
12     end Chraneno_Stavem;
```

- možnost použití atributů
- atribut `E'Count` vrátí počet zablokovaných vláken na vstupu do `entry E`
- např. implementace bariér
- funguje bezpečně pouze u chráněných objektů, nikoli tasků

Problém řízení přístupu ke zdrojům

- Aspekty synchronizace požadavků (Bloom, 1979)
 1. typ požadované služby
 2. pořadí požadavků
 3. interní stav přijímajícího vlákna
 4. priorita volajícího
 5. parametry požadavků
- Priorita: RT systémy
- Parametry: **requeue**

Problém řízení přístupu ke zdrojům

- Definice příkladu problému:
 - n zdrojů (vidliček v příborníku)
 - každé vlákno si může požadovat alokaci $1 \dots m$ zdrojů (vidliček) kde $m \leq n$
- Problém alokace více zdrojů najednou:
 - musím vstoupit do procedury, abych ověřil dostupnost zdrojů
- Možné řešení:
 - pollování (zodpovědnost vlákna)
 - rodiny entries (entry families) pro malé m
 - podpora přístupu k `in` parametrům předávaným v rámci volání rendezvous – možnost jejich použití ve stráži
 - ◆ není v Adě podporováno
 - ◆ implementováno např. v jazce SR (Synchronizing Resources)
 - ◆ problém s efektivitou implementace (bariéra se musí vyhodnocovat při každém řazení vlákna do fronty entry, nikoli jen jednou per entry)
 - `requeue`

Rodiny entries

- Úplná formální signatura entry

```
accept Entry_Name(Family_Index) (P : Parameters) do
2  -- sequence of statements
  exception
4  -- exception handling part
end Entry_Name;
```

- Rodiny entries
 - např. prioritizace a odlišení volajících vláken

```
task Multiplexer is
2   entry Channel(1..3) (X : in Data);
end Multiplexer;
```

Rodiny entries

- Příklad použití s vláknem
 - multiplexer vynucuje pořadí vstupů cyklicky 1, 2, 3

```
1 task body Multiplexer is
begin
3   loop
      for I in 1..3 loop
5         accept Channel(I) (X : in Data) do
              -- consume input data on channel I
7           end Channel;
      end loop;
9   end loop;
end Multiplexer;
```


Rodiny entries

- Příklad řízení zdrojů

```


1  type Request_Range is range 1 .. Max;
3  protected Resource_Controller is
4      entry Allocate(Request_Range) (R : out Resource);
5      procedure Release(R : Resource; Amount : Request_Range);
6  private
7      Free : Request_Range := Request_Range'Last;
8  end Resource_Controller;
9
10 protected body Resource_Controller is
11     entry Allocate(for F in Request_Range) (R : out Resource)
12         when F <= Free is
13     begin
14         Free := Free - F;
15     end Allocate;
16     procedure Release(R : Resource; Amount : Request_Range) is
17     begin
18         Free := Free + Amount;
19     end Release;
20 end Resource_Controller;

```


Rodiny entries

- Příklad řízení zdrojů
- Problémy
 - nevhodné pro větší množství alokovatelných zdrojů v jednom požadavku (m)
 - v případě soutěžení je výběr náhodný (prioritu lze nastavovat v rámci Real-Time Systems Annex)

„Eggshell“ model volání chráněného objektu




„Eggshell“ model volání chráněného objektu



- Přednost entry calls čekajících ve frontě
- Zámek vnější slupky
 - nedovolí vstup do slupky, pokud je jiné vlákno aktivní voláním procedury nebo entry
 - po vstupu se vyhodnocuje asociovaná podmínka (stráž)
 - tento mechanismus zajišťuje bezpečnost použití 'count'

„Eggshell“ model volání chráněného objektu



- Vnitřní slupka
 - po výstupu z každé procedury a entry se vyhodnocují podmínky a eventuálně se propouští volání čekající na vnitřní slupce

requeue

- Jak poslat za dveře někoho, koho už jsme si pustili do místnosti?
- **requeue** umožňuje vysunout aktivní vlákno v chráněném objektu do fronty před vnitřní slupku
 - nové volání musí mít stejnou signaturu
 - implicitně není přerušitelné pomocí **abort**, aby objekt nezůstal v rozpracovaném stavu
 - nové volání může být přerušitelné **requeue with abort**
- **requeue** umí fungovat i napříč více chráněnými objekty/úlohami
 - neobvyklé, používat opatrně

requeue

```
2 type Request_Range is range 1 .. Max;
3
4 protected Resource_Controller is
5     entry Allocate(R : out Resource; Amount : Request_Range);
6     procedure Release(R : Resource; Amount : Request_Range);
7 private
8     entry Assign(R : out Resource; Amount : Request_Range);
9     Free : Request_Range := Request_Range'Last;
10    New_Resources_Released : Boolean := False;
11    To_Try : Natural := 0;
12    ...
13 end Resource_Controller;
14
15 protected body Resource_Controller is
16     entry Allocate(R : out Resource; Amount : Request_Range)
17         when Free > 0 is
18     begin
19         if Amount <= Free then
20             Free := Free - Amount;
21             -- allocate
22         else
23             requeue Assign;
24         end if;
25     end Allocate;
```

requeue

```
2  entry Assign(R : out Resource; Amount : Request_Range)
   when New_Resources_Released is
3  begin
4      To_Try := To_Try - 1;
5      if To_Try = 0 then
6          New_Resources_Released := False;
7      end if;
8      if Amount <= Free then
9          Free := Free - Amount;
10         -- allocate
11     else
12         requeue Assign;
13     end if;
14 end Assign;
```



requeue

```
2  procedure Release(R : Resource; Amount : Request_Range) is
3  begin
4      Free := Free + Amount;
5      -- free resources
6      if Assign'Count > 0 then
7          To_Try := Assign'Count;
8          New_Resources_Released := True;
9      end if;
10     end Release;
11 end Resource_Controller;
```

Ada: Tasks, Rendezvous

- Koncept CSP: Communicating Sequential Processes
 - Hoare, 1978
 - paralelně běžící sekvenční procesy
 - komunikace: zasílání zpráv
 - synchronizace: synchronní zasílání zpráv
 - ◆ odesílatel se zablokuje, dokud příjemce není schopen přijmout zprávu
 - ◆ příjemce se zablokuje, dokud není schopen od odesílatele přijmout zprávu

Tasks, Rendezvous



Tasks

- **task**
 - lokálně definované
 - ◆ běží od začátku rozsahu, v němž jsou definované
 - dynamicky alokované
 - ◆ **access** typ
 - ◆ alokace pomocí **new**
 - ◆ běží až od alokace
 - pole tasků
- ukončování
 - spontánní
 - **abort**

Tasks

```
task T is
2 end T;

4 task body T is
begin
6     makam;
end T;

8
10 task type T_Type is
end T;

12 task body T_Type is
begin
14     loop
        makam;
16     end loop;
end T_Type;

18
20 Pole_T : array (1..10) of T_Type;

22 type T_Type_Access is access T_Type;
Dynamicky_T : T_Type_Access;
Dynamicky_T := new T_Type;
```

Parametrizace vláken

- předávání parametrů při vzniku vlákna
- užitečné s typy vláken

```
1 type Monitor_Procedure_Type is access procedure;  
  
3 task type Monitor_Task_Type (Mon_Proc : Monitor_Procedure_Type) is  
   entry Run;  
   entry Stop;  
   entry Request_Terminate;  
7 end Monitor_Task_Type;
```

Parametrizace vláken

```
1 task body Monitor_Task_Type is
    Finish_Flag : Boolean := False;
3    Terminate_Flag : Boolean := False;
begin
5    while not Terminate_Flag
        loop
7        select
            accept Run;
9            while not (Finish_Flag or Terminate_Flag)
                loop
11                 select
                    accept Stop do
13                     Finish_Flag := True;
                        end Stop;
15                 else
                    Mon_Proc.all;
17                 end select;
                end loop;
19            or
                accept Request_Terminate do
21                 Terminate_Flag := True;
                    end Request_Terminate;
23            end select;
                Finish_Flag := False;
25            end loop;
end Monitor_Task_Type;
```

Rendezvous

- místa synchronizace – předávání dat
- **entry**
 - deklarace rendezvous bodu
 - **in, out, in out** parametry
- **accept**
 - implementace v těle tasku

Tasks, Rendezvous

```

1 procedure Task1 is
2
3     task Vlakno is
4         entry ZadejX (X : in Integer);
5         entry PrectiX (X : out Integer);
6     end Vlakno;
7
8     task body Vlakno is
9         Hodnota : Integer;
10    begin
11        accept ZadejX (X : in Integer) do
12            Hodnota := X;
13        end ZadejX;
14        Hodnota := Hodnota + 1;
15        accept PrectiX (X : out Integer) do
16            X := Hodnota;
17        end PrectiX;
18    end Vlakno;
19
20    Chci_Inkrementovat : Integer;
21
22 begin
23     Vlakno.ZadejX(Chci_Inkrementovat);
24     Vlakno.PrectiX(Chci_Inkrementovat);
25 end Task1;

```

Rendezvous

- **select**

- výběr z více **accept** možností

```
1  task body T is
begin
3    loop
      select
5        accept Rande1 do
          neco;
7        end Rande1;
      or
9        accept Rande2 do
          neco;
11       end Rande2;
          neco;
13       accept Rande3 do
          neco;
15       end Rande3;
      or
17       terminate;
    end loop;
19 end T;
```

Rendezvous

- **select**

- časovaný výběr

```
1 task body T is
begin
3   loop
   select
5       accept Randel1 do
           neco;
7       end Randel1;
   or
9       delay 10.0;
           taky_neco;
11      end select;
   end loop;
13 end T;
```

Rendezvous

- **select**

- časovaný výběr

```
1 task body T is
begin
3   loop
   select
5       accept Randel do
           neco;
7       end Randel;
   else -- ekvivalent "or delay 0.0"
9       null; -- busy waiting
   end select;
11  end loop;
end T;
```

Rendezvous

- výjimky během rendezvous
 - jsou doručeny jak volajícímu vláknu, tak i vlastnímu vláknu
 - pokud výjimka není ošetřena ve vláknu, je vlákno ukončeno a výjimka se nepropaguje do rodiče (považováno za příliš disruptivní)

```
begin
2   select
      accept Volani do
4       ...
      raise Chyba;
6       ...
      end Volani;
8   end select;
exception
10  when Chyba =>
      Naprav_Stav;
12  when other =>
      Oznam_Uzivateli;
14 end;
```

Rendezvous

- výjimky během rendezvous
 - jsou doručeny jak volajícímu vláknu, tak i vlastnímu vláknu
 - pokud výjimka není ošetřena ve vláknu, je vlákno ukončeno a výjimka se nepropaguje do rodiče (považováno za příliš disruptivní)

```
---  
2 ---  
4 begin  
   T.Volani;  
6 exception  
   when Chyba =>  
8     Oznam_Uzivateli;  
end;
```

Chráněné entries u vláken

- podobně jako u chráněných objektů s mírně odlišnou syntaxí

```

1 task body Ukazka is
2   Zinicializovano : Boolean := False;
3   Hodnota : Data;
4 begin
5   loop
6     select
7       when Zinicializovano =>
8         accept Cti (H : out Data) do
9           H := Hodnota;
10        end;
11    or
12    accept Zapis (H : in Data) do
13      Hodnota := H;
14    end;
15    Zinicializovano := True;
16  end select;
17 end loop;
18 end Ukazka;

```

- podmínka se vyhodnocuje při každém průchodu přes **select**
- pokud není žádná podmínka splněna, je vyhozena výjimka **Program_Error** (možno použít strukturu **select ... else ... end select;**)
- změna hodnot mezi testem a rendezvous (viz komentář u 'Count')

Atributy vláken

- Ada 95 Quality and Style Guide, Section 6.2.3
- ' **Terminated**
 - bezpečné pouze testování na **True** (po ukončení vlákno nemůže obživnout)
- ' **Callable**
 - bezpečné pouze testování na **False** (po ukončení vlákno nemůže obživnout)


Atributy vláken

- Ada 95 Quality and Style Guide, Section 6.2.3
- 'Count
 - chování vlákna by nemělo záviset na tomto atributu (používat raději jen s chráněnými objekty)

```

1 select
2   when Transmit'Count > 0 and Receive'Count = 0 =>
3     accept Transmit;
4   ...
5   or
6   accept Receive;
7   ...
8 end select;

```



stav 'Count se může změnit mezi vyhodnocením a následnou akcí (např. volající použil časově omezené volání a mezi testem a `accept` se ukončil)

- u chráněných objektů: každá práce s frontou je chráněná (viz eggshell model)

Chráněné entries s timeoutem

- Nelze implementovat pomocí chráněných typů, jsou třeba vlákna

```
task body Ukazka is
2   Zinicializovano : Boolean := False;
   Hodnota : Data;
4   begin
   loop
6     select
       when Zinicializovano =>
8       accept Cti (H : out Data) do
           H := Hodnota;
10      end;
       or
12      delay 1.0;
           -- neco
14     end select;
   end loop;
16 end Ukazka;
```

Vynucené ukončování vláken

- Alternativní příklad

```

select
2   T.Ukonci;
   or
4   delay 180*Seconds;
   abort T;
6 end select;

```

nebo pokud nevěříme, že se úloha po `T.Ukonci` ukončí

```

select
2   T.Ukonci;
   delay 60*Seconds;
4  or
   delay 180*Seconds;
6 end select;
  abort T;

```

- pozor na nebezpečí použití `abort`
- `pragma Restrictions (No_Abort_Statements);`
- ani druhé řešení není blbuvzdorné (pokud se `T.Ukonci` může zakousnout v rámci bloku `accept`, použití `requeue`)

Asynchronous Transfer of Control

- Potřeba *rychle* reagovat na asynchronní události
 - reakce na chyby (např. výpadek HW, kvůli němuž se akce nikdy nedokončí)
 - změny režimů v důsledku (neočekávaných) událostí
 - dosažení co nejlepšího výsledku v případě iterativního přerušitelného výpočtu
 - přerušení uživatelem

Asynchronous Transfer of Control

• Struktura

```

select
2   -- triggering_statement
   delay 5.0;
4   -- post_trigger_part
   Put_Line ("Tudy cesta nevede!");
6 then abort
   -- abortable_part
8   Prevelevelmidlouhe_Volani;
end select;

```

- pokud `abortable_part` doběhne dříve než `triggering_statement`, pokusí se ukončit `triggering_statement`
- pokud `triggering_statement` doběhne dřív než `abortable_part`, je `abortable_part` ukončena a provede se část `post_trigger_part`
- `triggering_statement` – v Ada 95 `delay/entry`, v Ada 2005 i procedury
- `abortable_part` nemusí být implementována jako samostatný task

Asynchronous Transfer of Control

- Příklad: iterativní dlouhý výpočet, chceme nejlepší odhad v době, kdy jej potřebujeme (J. Barnes, Ada 95)
 - chráněný objekt na předávání posledního výsledku

```
1 begin Result is
    procedure Set_Estimate(X : in Data);
3     function Get_Estimate return Data;
    private
5     Est : Data;
end;
```

- signalizační objekt (např. uživatel chce výsledek)

```
begin Trigger is
2     entry Wait;
    -- when Flag
4     procedure Signal;
    private
6     Flag : Boolean := False;
end;
```

Asynchronous Transfer of Control

- Příklad: iterativní dlouhý výpočet, chceme nejlepší odhad v době, kdy jej potřebujeme (J. Barnes, Ada 95)

- použití

```
1 select
    Trigger.Wait;
3 then abort
    Computation;
5 end select;
```

```
1 Trigger.Signal;
   E := Result.Get_Estimate;
```

- oddělení logiky výpočtu od jeho ukončování – výpočet nemusí zjišťovat, kdy má končit

Asynchronous Transfer of Control

- Výjimky při ATC
 - pokud se odehraje jen jedna výjimka (v jedné z částí), je možno ji zachytit
 - pokud se odehrají dvě výjimky současně v řídicí i přerušitelné části, je výjimka z přerušitelné části ztracena

Earliest Deadline First Dispatching

- Komplikovanější koncept aktivní priority
 - pokud nějaké vlákno B pracuje v chráněném objektu (tedy s ceiling prioriton P) a vlákno A má dřívější termín, je vlákno A zařazeno do fronty s prioritou větší než P (existuje-li taková fronta)
 - pokud nikdo nepracuje s chráněnými objekty, je vlákno A zařazeno do fronty s prioritou **Priority'First**
 - vlákno A podědí aktivní prioritu fronty
- Dispatching points pro vlákno A při použití EDF
 - změna termínu A
 - zkrácení termínu pro úlohu B ve frontě s prioritou A, pokud nový termín B je nastane dříve jako termín A
 - pokud se objeví úloha ve frontě s prioritou vyšší než A
- Problém s implementovatelností na běžných OS (poznámka ve specifikaci balíku v GNATu)
 - implementováno např. pro MARTE OS (<http://marte.unican.es/>)

Zpoždění vzbuzení

- Vzbuzení po použití `delay` a `delay until` je nejdříve se specifikovaném okamžiku, ale může nastat i později
 - zpoždění se označuje jako *lateness*
- Použití `delay` a `delay until` má nějakou režii i v případě, že se fakticky nečeká (tj. parametry vyustí v `delay 0.0`)
 - režie se promítá do kódu v případě specifikace `dispatching points` pomocí `delay 0.0`

Zpoždění vzbuzení

- Příklad pro GNATforLEON <http://polaris.dit.upm.es/~ork/download/opm-2.1.0.pdf>
- *The implementation shall document the following metrics (only those metrics that are significant in the context of the Ravenscar profile are cited):*
 - *An upper bound on the lateness of a **delay until** statement, in a situation where the value of the requested expiration time is after the time the task begins executing the statement, the task has sufficient priority to preempt the processor as soon as it becomes ready, and it does not need to wait for any other execution resources. The upper bound is expressed as a function of the difference between the requested expiration time and the clock value at the time the statement begins execution. The lateness of a **delay until** statement is obtained by subtracting the requested expiration time from the real time that the task resumes execution following this statement.*

The following measurements have been performed:

- ◆ *One task + background task* The delay until lateness upper bound for a call to a delay until statement is 8051 clock cycles (161 μ s), using a 50 MHz system clock. This lateness occurs when the time of the delay until coincides with a second boundary. It must be noted that the clock interrupt occurs every second in the kernel tested. If the time of the delay until statement does not coincide with a clock interrupt, the lateness upper bound for the execution of a delay until statement is 7061 clock cycles (141.2 μ s).

Hodiny reálného času

- **Ada.Real_Time** – monotónní hodiny s vysokým rozlišením
 - **Time** – vyjádření časového okamžiku
 - **Time_Span** – vyjádření rozsahu trvání/intervalu
 - srovnání (minimálních) požadavků na hodiny v Adě

	Calendar	Real_Time
rozsah času	500 let	50 let
rozsah intervalu	1 den	± 1 hodina
přesnost	20 ms	20 μ s

Rozsah **Real_Time** může být menší na platformách se slovem kratším jak 32 b.

```

with Ada.Real_Time; use Ada.Real_Time;
2
begin
4   T_One : Time := Clock;
   TS : Time_Span := To_Time_Span(1.0);
6   T_Two : Time := T_One + TS;
   delay until T_Two;
8   delay To_Duration (TS);
end;
```

Časovače událostí

- `Ada.Real_Time.Timing_Events`
- Využití pro událostmi řízené programování bez vláken a komplikace kódu pomocí `delay`

```
package Ada.Real_Time.Timing_Events is
2
   type Timing_Event is tagged limited private;
4
   type Timing_Event_Handler
6     is access protected procedure (Event : in out Timing_Event);
8
   procedure Set_Handler
       (Event   : in out Timing_Event;
        At_Time : Time;
        Handler : Timing_Event_Handler);
10
   procedure Set_Handler
12     (Event   : in out Timing_Event;
        In_Time : Time_Span;
        Handler : Timing_Event_Handler);
14
   function Current_Handler
16     (Event : Timing_Event) return Timing_Event_Handler;
18
   procedure Cancel_Handler
20     (Event       : in out Timing_Event;
        Cancelled  : out Boolean);
22
   function Time_Of_Event (Event : Timing_Event) return Time;
24
```

Časovače událostí

- **Timing_Event**
 - tagovaný typ – možnost rozšíření o vlastní data
 - privátní ne-abstraktní typ – nepotřebuje run-time dispatching
- **Timing_Event_Handler**
 - obdoba obsluhy přerušení
 - ukazatel na **access protected procedure**
- **Set_Handler**
 - varianta s absolutním časem
 - varianta s relativním časem
 - **null** místo ukazatele na proceduru vymaže časovač (ekvivalent **Cancel_Handler**)
 - opakované volání přepisuje budoucí událost
- Spuštění události
 - co nejdříve poté, co uplyne specifikovaný čas
 - z důvodů efektivity se obvykle pověsí na přerušení hodin ≤ 10 ms (obdobně jako ošetřování **delay** ve vláknech)
- Nevýhoda: kód běží s prioritou **Interrupt_Priority**
 - pro složitější a déle běžící úlohy je lépe používat vlákna
 - míněno jako lightweight mechanismus pro omezené platformy

Odlehčená komunikace mezi vlákny

- Synchronní komunikace mezi vlákny

```

package Ada.Synchronous_Task_Control is
2
  type Suspension_Object is limited private;
4  procedure Set_True (S : in out Suspension_Object);
  procedure Set_False (S : in out Suspension_Object);
6  function Current_State (S : Suspension_Object) return Boolean;
  procedure Suspend_Until_True (S : in out Suspension_Object);

```

- ekvivalent `wait/notify`
- `Set_True`, `Set_False`, `Current_State` jsou vzájemně atomické a neblokující
- `Suspend_Until_True` přepoklopí `suspension object` zpět na `False`

Odlehčená komunikace mezi vlákny

• Asynchronní komunikace mezi vlákny

```
1 package Ada.Asynchronous_Task_Control is
2
3     pragma Unimplemented_Unit;
4     procedure Hold (T : Ada.Task_Identification.Task_Id);
5     procedure Continue (T : Ada.Task_Identification.Task_Id);
6     function Is_Held (T : Ada.Task_Identification.Task_Id) return Boolean;
7
8 end Ada.Asynchronous_Task_Control;
```

- umožňuje zasuspendovat jiné vlákno – potenciálně nebezpečné
- koncept idle task priority
- suspendování se provádí pomocí snížení priority pod idle task priority
 - ◆ volání `Hold` na vlákno řízené EDF jej dočasně vyloučí z EDF
 - ◆ dispatching points odpovídají plánovači, kterým jsou vlákna v daném okamžiku řízena
 - ◆ řeší problém, aby se vlákno nezasuspendovalo uvnitř chráněného objektu (nejsou v něm dispatching points)
 - ◆ pokud je zavolán `accept` zasuspendovaného vlákna, je vykonán, protože podědí prioritu volajícího
 - ◆ pokud je vlákno blokováno uvnitř chráněného objektu v čekání na otevření strážce entry, je uvolněno, pokud je se stráž otevře a vlákno je jediné ve frontě

Možnosti omezení

- **`pragma Restrictions`** – kontrolované před během programu
 - `No_Dynamic_Priorities`** There is no use of dynamic priorities.
 - `No_Dynamic_Attachments`** There are no calls to any of the operations defined in package Interrupts, e.g. Attach Handler.
 - `No_Local_Protected_Objects`** Protected objects are only declared at the library level.
 - `No_Local_Timing_Events`** Timing events are only declared at the library level.
 - `No_Protected_Type_Allocators`** There are no allocators for protected types or types containing protected subcomponents.
 - `No_Relative_Delay`** There are no relative delay statements (i.e. delay).
 - `No_Queue_Statements`** There are no queue statements.
 - `No_Select_Statements`** There are no select statements.
 - `No_Specific_Termination_Handlers`** There are no calls to the specific handler routines in the task termination package.
 - `Simple_Barriers`** The boolean expression in an entry barrier is either a static boolean expression or a boolean component of the enclosing protected object (e.g. a simple boolean variable).

Možnosti omezení

- **pragma Restrictions** – nedefinované místo kontroly

Max_Select_Alternatives Specifies the maximum number of alternatives in a select statement.

Max_Task_Entries Specifies the maximum number of entries per task. The maximum number of entries for each task type (including those with entry families) must be determinable at compile-time. A value of zero indicates that no rendezvous is possible.

Max_Protected_Entries Specifies the maximum number of entries per protected type. The maximum number of entries for each protected type (including those with entry families) must be determinable at compile-time.

Možnosti omezení

- **pragma Restrictions** – kontrola za běhu

No_Task_Termination All tasks are non-terminating. It is implementation defined what happens if a task terminates – but any fall-back handler must be executed as the first task terminates.

Max_Storage_At_Blocking Specifies the maximum portion (in storage elements) of a task's storage size that can be retained by a blocked task. If a check fails, Storage Error is raised at the point where the respective construct is elaborated.

Max_Asynchronous_Select_Nesting Specifies the maximum dynamic nesting level of asynchronous select statements. A value of zero prevents the use of any such statement. If a check fails, Storage Error is raised as above.

Max_Tasks Specifies the maximum number of tasks, excluding the environment task, that are allowed to exist over the lifetime of a partition. A zero value prevents tasks from being created. If a check fails, Storage Error is raised as above.

Max_Entry_Queue_Length This defines the maximum number of calls queued on an entry. Violation will cause Program Error to be raised at the point of call.

Ravenscar

- `pragma Profile (Ravenscar);`

```
1 pragma Task_Dispatching_Policy (FIFO_Within_Priorities);  
2 pragma Locking_Policy (Ceiling_Locking);  
3 pragma Detect_Blocking;  
4 pragma Restrictions (  
5         No_Abort_Statements,  
6         No_Dynamic_Attachment,  
7         No_Dynamic_Priorities,  
8         No_Implicit_Heap_Allocations,  
9         No_Local_Protected_Objects,  
10        No_Local_Timing_Events,  
11        No_Protected_Type_Allocators,  
12        No_Relative_Delay,  
13        No_Requeue_Statements,  
14        No_Select_Statements,  
15        No_Specific_Termination_Handlers,  
16        No_Task_Allocators,  
17        No_Task_Hierarchy,  
18        No_Task_Termination,  
19        Simple_Barriers,  
20        Max_Entry_Queue_Length => 1,  
21        Max_Protected_Entries => 1,  
22        Max_Task_Entries => 0,  
23        No_Dependence => Ada.Asynchronous_Task_Control,  
24        No_Dependence => Ada.Calendar,  
25        No_Dependence => Ada.Execution_Time.Group_Budget,  
26        No_Dependence => Ada.Execution_Time.Timers,  
27        No_Dependence => Ada.Task_Attributes);
```

Ravenscar

- Nesmí být hierarchie vláken
 - vlákna se musí být deklarována na úrovni knihoven, nikoli z hlavního vlákna
- Zákaz rendezvous
 - vlákna se musí synchronizovat přes chráněné objekty
- Předávání dat
 - atomické proměnné
 - chráněné objekty
 - využití suspension objects
- ```
Ada.Synchronous_Task_Control.Suspend_Until_True(S);
Ada.Synchronous_Task_Control.Set_True(Periodic.S);
```
- Omezení na nejvýše jedno entry per chráněný objekt/typ
  - kombinace protected type a suspension object
- Pouze jedno vlákno smí čekat ve vnitřní slupce entry
  - separátní entries pro různá vlákna

# Ravenscar – příklady

- Periodická úloha

```
1 task type Periodicka (Prio : System.Priority; Cyklus : Positive) is
2 pragma Priority (Prio);
3 end Periodicka;
4
5 task body Periodicka is
6 Dalsi_Cas : Ada.Real_Time.Time;
7 Interval : constant Ada.Real_Time.Time_Span :=
8 Ada.Real_Time.Microseconds (Cyklus);
9 begin
10 Dalsi_Cas := Ada.Real_Time.Clock + Interval;
11 loop
12 -- neco
13 delay until Dalsi_Cas;
14 Dalsi_Cas := Dalsi_Cas + Interval;
15 end loop;
16 end Periodicka;
```







# Ravenscar – příklady

- Ravenscar verze
  - zapisující vlákna se zastavují na suspension objektu

```

1 task body Probe_Collector (ID: Probe_ID) is
2 begin
3 Ada.Synchronous_Task_Control.Set_True(S(ID));
4 loop
5 Ada.Synchronous_Task_Control.Suspend_Until_True(S(ID));
6 delay until Next;
7 ...
8 Probe_Protector_Ravenscar.Write(D, ID);
9 Next := Next + Interval;
10 end loop;
11 end Probe_Collector;

```

Zdroj: M. Ben-Ari, Ada for Software Engineers, 2nd ed. for Ada 2005